	* March 24, 2016
	* Hannes Schwandt, hannes.schwandt@uzh.ch

	
	*********************************************************************
	*----------------------- Define Directories ------------------------*
	*********************************************************************

	global data_cleaned 		"/data/"
	cd 		"/results/"
	
	
	**********************************************
	*  Figure 1: Life expectancy at birth by poverty percentile and gender *
	**********************************************
	
		***********************
		*  Figure 1 A: Levels *
		***********************
		
		use ${data_cleaned}LE_quantile, clear
		keep if q1990==0

		twoway ///
		(scatter LEmale quantile if year==1990, mcolor(green) msymbol(T) ) ///  
		(lfit LEmale quantile if year==1990, lcolor(green) ) ///  
		(lfit LEmale quantile if year==2000, lcolor(green) lpattern(##-) ) ///  
		(scatter LEmale quantile if year==2010, mcolor(green) msymbol(h) ) ///  
		(lfit LEmale quantile if year==2010, lcolor(green) ) ///  
		(scatter LEfemale quantile if year==1990, mcolor(blue) msymbol(Th) ) ///  
		(lfit LEfemale quantile if year==1990, lcolor(blue)  ) ///  
		(lfit LEfemale quantile if year==2000, lcolor(blue) lpattern(##-) ) ///  
		(scatter LEfemale quantile if year==2010, mcolor(blue) msymbol(Oh) ) ///  
		(lfit LEfemale quantile if year==2010, lcolor(blue)   ) ///  
		,  subtitle("(A) Level", size(4)) graphregion(color(white)) ///
		graphregion(color(white)) ylabel(#3)  xtitle("Poverty percentile", height(4)) ytitle("Life expectancy at birth")   ///
		 legend(order(1 "Men 1990" 3 "M 2000" 4 "M 2010" 6 "Women 1990" 8 "W 2010" 9 "W 2010") symxsize(5) symysize(.2) row(2) size(3)) saving(LE1, replace) nodraw
	
		*********************************
		*  Figure 1 B: Change 2010-1990 *
		*********************************
				
			use ${data_cleaned}LE_quantile, clear
			drop if year==2000
			keep if q1990==0

			bysort quantile  (year): gen dmale=(LEmale[_n+1]-LEmale) if year==1990
			bysort quantile  (year): gen dfemale=(LEfemale[_n+1]-LEfemale)  if year==1990
					
			twoway ///
			(scatter dmale quant , mcolor(green) msymbol(S) ) ///  
			(lfit dmale quant if year==1990, lcolor(green) ) ///  
			(scatter dfemale quant , mcolor(blue) msymbol(Sh) ) ///  
			(lfit dfemale quant if year==1990, lcolor(blue) ) ///  
			, subtitle("(B) Change", size(4)) graphregion(color(white)) ylabel(#3)  xtitle("Poverty percentile", height(4)) ytitle("Change in life expectancy at birth")   ///
			saving(LE2, replace)  ///
			 legend(order(1 "Change Men" 3 "Change Women") symxsize(3) symysize(.2) row(2) size(3)) nodraw
			
		graph combine LE1.gph LE2.gph, graphregion(color(white)) saving(Fig1, replace)

			*slope of regression line
			reg dmale quant 
			reg dfemale quant 
	
	
	
	********************************************************************************************
	*  Figures 2/3: Male/Female 3-year mortality rates by poverty percentile across age groups *
	********************************************************************************************
	
			
	use ${data_cleaned}mort_quantiles_county, clear
	
	keep if qtype=="Poverty quantile" & q1990==0
	gen mort=(deaths/pop)*1000

	foreach g of numlist   0 1 {
	foreach a of numlist  0 2 7 12 17 22 27 32 37 42 47 52 57 62 67 72 77 82 90{
	
	#delimit; 
	local gender_0 "women" ; local gender_1 "men" ;
	local age_0 "0" ; local age_2 "1-4" ; local age_7 "5-9" ; local age_12 "10-14" ; local age_17 "15-19" ; local age_22 "20-24" ; local age_27 "25-29" ; local age_32 "30-34" ; local age_37 "35-39" ; local age_42 "40-44" ; local age_47 "45-49" ; local age_52 "50-54" ; local age_57 "55-59" ; local age_62 "60-64" ; local age_62 "60-64" ; local age_67 "65-69" ; local age_72 "70-74" ; local age_77 "75-79" ; local age_82 "80-84" ; local age_90 "85 and above" ;
	local ytitle_0 "3yr mortality [per 1,000]"; local ytitle_22 "3yr mortality [per 1,000]"; local ytitle_47 "3yr mortality [per 1,000]"; local ytitle_72 "3yr mortality [per 1,000]"; 
	#delimit cr
		
	twoway ///
	(scatter mort quantile if year==1990, mcolor(blue) msymbol(T) ) /// 
	(lfit mort quantile if year==1990, lcolor(blue) ) ///  
	(lfit mort quantile if year==2000, lcolor(black) lwidth(.5) lpattern(#-#)) ///  
	(scatter mort quantile if year==2010, mcolor(green) msymbol(Oh) ) ///  
	(lfit mort quantile if year==2010, lcolor(green) ) ///  
	if q1990==0 & age==`a' &  male==`g',  ///
	subtitle("Age `age_`a'' ", size(5)) graphregion(color(white)) ylabel(#3)  xtitle("Poverty percentile", height(2)) ytitle("`ytitle_`a''")   ///
	legend(order(1 "1990" 3 "2000" 4 "2010") symxsize(4) symysize(.2) row(1) size(2)) saving("`gender_`g''_`age_`a''", replace) nodraw
	}
	
	local f=3-`g'
	
	grc1leg "`gender_`g''_`age_0'.gph" "`gender_`g''_`age_2'.gph" "`gender_`g''_`age_7'.gph" "`gender_`g''_`age_12'.gph" "`gender_`g''_`age_17'.gph" "`gender_`g''_`age_22'.gph" "`gender_`g''_`age_27'.gph" "`gender_`g''_`age_32'.gph" "`gender_`g''_`age_37'.gph" "`gender_`g''_`age_42'.gph" "`gender_`g''_`age_47'.gph" "`gender_`g''_`age_52'.gph" "`gender_`g''_`age_57'.gph" "`gender_`g''_`age_62'.gph" "`gender_`g''_`age_67'.gph" "`gender_`g''_`age_72'.gph" "`gender_`g''_`age_77'.gph" "`gender_`g''_`age_82'.gph" "`gender_`g''_`age_90'.gph"  ///
	, graphregion(color(white)) legendfrom("`gender_`g''_`age_0'.gph") ///
	saving("Fig`f'", replace)	
	}
	
			foreach g of numlist   0 1 {
			foreach a of numlist  0 2 7 12 17 22 27 32 37 42 47 52 57 62 67 72 77 82 90{
			erase "`gender_`g''_`age_`a''.gph"		
			}
			}
			
		

	*************************************************************************************
	*  Figure S1: Population of county groups ordered by poverty rate, in 1990 and 2010 *
	*************************************************************************************
	
		use ${data_cleaned}mort_quantiles_county, clear					
		keep if qtype=="Poverty quantile" & q1990==0

		collapse  (rawsum) pop , by(quantile year)

		
		replace pop=pop/1000000
						
		twoway  (scatter pop quant if year==1990, mcolor(blue) ) ///
				(scatter pop quant if year==2010, mcolor(green) msymbol(Oh) ), ///
				legend(order(1 "1990" 2 "2010")) graphregion(color(white)) ///
				ytitle("Population (in millions)") xtitle("Poverty percentile") ylabel(0(5)20) saving(FigS1, replace) 
				
	
	
		*************************************************************************************************
		*  Figure S2: 3-year mortality rates ranked using alternative county characteristics, age 0-19  *
		*************************************************************************************************
		
			{
			
			use ${data_cleaned}mort_quantiles_county, clear
			
			keep if inrange(age,0 ,19)
			
			collapse (sum) deaths pop, by(quant qtype q1990 male year)
			
			gen mort=deaths/pop*1000

			local z=0
			foreach name in "Poverty quantile" "HS dropout quantile" "Median income quantile"  "LE quantile" {
			local z=`z'+1
			
			local Name_1 "Poverty percentile" 
			local Name_3 "Median income percentile" 
			local Name_2 "High school dropout percentile" 
			local Name_4 "Life expectancy percentile" 
			
					foreach g of numlist 1 0 {
					
					local a=1
					#delimit; 
					local gender_0 "Women" ; local gender_1 "Men" ;
					local ytitle_1_1 "3yr mortality [per 1,000]"; local ytitle_3_1 "3yr mortality [per 1,000]" ; 
					#delimit cr
					
					
					twoway ///
					(scatter mort quantile if year==1990, mcolor(blue) msymbol(T) ) ///  
					(lfit mort quantile if year==1990, lcolor(blue) ) ///  
					(lfit mort quantile if year==2000, lcolor(black) lwidth(.2) lpattern(#-)) ///  
					(scatter mort quantile if year==2010, mcolor(green) msymbol(Oh) ) ///  
					(lfit mort quantile if year==2010, lcolor(green) ) ///  
					if q1990==0 &  male==`g' & qtype=="`name'",  ///
					subtitle("`gender_`g''") graphregion(color(white)) ylabel(#3)  xtitle("Percentile", height(4)) ytitle("`ytitle_`a'_`g''")   ///
					legend(order(1 "1990" 3 "2000" 4 "2010") symxsize(3) symysize(.2) row(1) size(2)) saving("`gender_`g''_`z'", replace) nodraw ///
					ylabel(0(2)4)
					}
					
					grc1leg "men_`z'.gph" "women_`z'.gph"  ///
					, graphregion(color(white)) legendfrom("men_`z'.gph") ///
					subtitle("`Name_`z''", size(3.5)) saving("ptype_`z'", replace) ycommon
					
					}
			
			grc1leg "ptype_1.gph" "ptype_2.gph" "ptype_3.gph" "ptype_4.gph" ///
			, graphregion(color(white))  legendfrom("ptype_1.gph") ///
			subtitle("Age 0-19", size(3)) saving("FigS2", replace) ycommon
			
			erase ptype_1.gph 
			erase ptype_2.gph 
			erase ptype_3.gph 
			erase ptype_4.gph 
			}
						
	
		*************************************************************************************************
		*  Figure S3: 3-year mortality rates ranked using alternative county characteristics, age 20-49 *
		*************************************************************************************************
	
				{
			
			use ${data_cleaned}mort_quantiles_county, clear
			
			keep if inrange(age,20 ,49)
			
			collapse (sum) deaths pop, by(quant qtype q1990 male year)
			
			gen mort=deaths/pop*1000

			local z=0
			foreach name in "Poverty quantile" "HS dropout quantile" "Median income quantile"  "LE quantile" {
			local z=`z'+1
			
			local Name_1 "Poverty percentile" 
			local Name_3 "Median income percentile" 
			local Name_2 "High school dropout percentile" 
			local Name_4 "Life expectancy percentile" 
			
					foreach g of numlist 1 0 {
					
					local a=1
					#delimit; 
					local gender_0 "Women" ; local gender_1 "Men" ;
					local ytitle_1_1 "3yr mortality [per 1,000]"; local ytitle_3_1 "3yr mortality [per 1,000]" ; 
					#delimit cr
					
					
					twoway ///
					(scatter mort quantile if year==1990, mcolor(blue) msymbol(T) ) ///  
					(lfit mort quantile if year==1990, lcolor(blue) ) ///  
					(lfit mort quantile if year==2000, lcolor(black) lwidth(.2) lpattern(#-)) ///  
					(scatter mort quantile if year==2010, mcolor(green) msymbol(Oh) ) ///  
					(lfit mort quantile if year==2010, lcolor(green) ) ///  
					if q1990==0 & male==`g' & qtype=="`name'",  ///
					subtitle("`gender_`g''") graphregion(color(white)) ylabel(#3)  xtitle("Percentile", height(4)) ytitle("`ytitle_`a'_`g''")   ///
					legend(order(1 "1990" 3 "2000" 4 "2010") symxsize(3) symysize(.2) row(1) size(2)) saving("`gender_`g''_`z'", replace) nodraw
					}
					
					grc1leg "men_`z'.gph" "women_`z'.gph"  ///
					, graphregion(color(white)) legendfrom("men_`z'.gph") ///
					subtitle("`Name_`z''", size(3.5)) saving("ptype_`z'", replace) ycommon
					
					}
			
			grc1leg "ptype_1.gph" "ptype_2.gph" "ptype_3.gph" "ptype_4.gph" ///
			, graphregion(color(white))  legendfrom("ptype_1.gph") ///
			subtitle("Age 20-49", size(3)) saving("FigS3", replace) ycommon
		
			erase ptype_1.gph 
			erase ptype_2.gph 
			erase ptype_3.gph 
			erase ptype_4.gph 
	
			}

	
		*************************************************************************************************
		*  Figure S4: 3-year mortality rates ranked using alternative county characteristics, age 50-84 *
		*************************************************************************************************
		
			{
			
			use ${data_cleaned}mort_quantiles_county, clear
			
			keep if inrange(age,50 ,84)
			
			collapse (sum) deaths pop, by(quant qtype q1990 male year)
			
			gen mort=deaths/pop*1000

			local z=0
			foreach name in "Poverty quantile" "HS dropout quantile" "Median income quantile"  "LE quantile" {
			local z=`z'+1
			
			local Name_1 "Poverty percentile" 
			local Name_3 "Median income percentile" 
			local Name_2 "High school dropout percentile" 
			local Name_4 "Life expectancy percentile" 
			
					foreach g of numlist 1 0 {
					
					#delimit; 
					local gender_0 "Women" ; local gender_1 "Men" ;
					local ytitle_1_1 "3yr mortality [per 1,000]"; local ytitle_3_1 "3yr mortality [per 1,000]" ; 
					#delimit cr
					
					
					twoway ///
					(scatter mort quantile if year==1990, mcolor(blue) msymbol(T) ) ///  
					(lfit mort quantile if year==1990, lcolor(blue) ) ///  
					(lfit mort quantile if year==2000, lcolor(black) lwidth(.2) lpattern(#-)) ///  
					(scatter mort quantile if year==2010, mcolor(green) msymbol(Oh) ) ///  
					(lfit mort quantile if year==2010, lcolor(green) ) ///  
					if q1990==0 &  male==`g' & qtype=="`name'",  ///
					subtitle("`gender_`g''") graphregion(color(white)) ylabel(#3)  xtitle("Percentile", height(4)) ytitle("`ytitle_`a'_`g''")   ///
					legend(order(1 "1990" 3 "2000" 4 "2010") symxsize(3) symysize(.2) row(1) size(2)) saving("`gender_`g''_`z'", replace) nodraw ///
					ylabel(20(40)100)
					}
					
					grc1leg "men_`z'.gph" "women_`z'.gph"  ///
					, graphregion(color(white)) legendfrom("men_`z'.gph") ///
					subtitle("`Name_`z''", size(3.5)) saving("ptype_`z'", replace) ycommon
					
					}
			
			grc1leg "ptype_1.gph" "ptype_2.gph" "ptype_3.gph" "ptype_4.gph" ///
			, graphregion(color(white))  legendfrom("ptype_1.gph") ///
			subtitle("Age 50-84", size(3)) saving("FigS4", replace) ycommon
			
			erase ptype_1.gph 
			erase ptype_2.gph 
			erase ptype_3.gph 
			erase ptype_4.gph 
	
			}

	
		*************************************************************************************************
		*  Figure S5: Life expectancy at birth across poverty percentiles, gender, and years, using life expectancy estimates provided by the Institute for Health Metrics and Evaluation (IHME) *
		*************************************************************************************************
	
			use ${data_cleaned}LE_quantile, clear
			keep if q1990==0

			
			twoway ///
			(scatter LEmale quantile if year==1990, mcolor(green) msymbol(T) ) ///  
			(lfit LEmale quantile if year==1990, lcolor(green) ) ///  
			(lfit LEmale quantile if year==2000, lcolor(green) lpattern(##-) ) ///  
			(scatter LEmale quantile if year==2010, mcolor(green) msymbol(h) ) ///  
			(lfit LEmale quantile if year==2010, lcolor(green) ) ///  
			(scatter LEfemale quantile if year==1990, mcolor(blue) msymbol(Th) ) ///  
			(lfit LEfemale quantile if year==1990, lcolor(blue)  ) ///  
			(lfit LEfemale quantile if year==2000, lcolor(blue) lpattern(##-) ) ///  
			(scatter LEfemale quantile if year==2010, mcolor(blue) msymbol(Oh) ) ///  
			(lfit LEfemale quantile if year==2010, lcolor(blue)   ) ///  
			,  subtitle("(A) Baseline life expectancy estimates", size(4)) graphregion(color(white)) ///
			graphregion(color(white)) ylabel(#3)  xtitle("Poverty percentile", height(4)) ytitle("Life expectancy at birth")   ///
			 legend(order(1 "Men 1990" 3 "Men 2000" 4 "Men 2010" 6 "Women 1990" 8 "Women 2010" 9 "Women 2010") symxsize(5) symysize(.2) row(2) size(3)) saving(LE1, replace) nodraw
			
			
			twoway ///
			(scatter IHME_LEmale quantile if year==1990, mcolor(green) msymbol(T) ) ///  
			(lfit IHME_LEmale quantile if year==1990, lcolor(green) ) ///  
			(scatter IHME_LEmale quantile if year==2010, mcolor(green) msymbol(h) ) ///  
			(lfit IHME_LEmale quantile if year==2010, lcolor(green) ) ///  
			(lfit IHME_LEmale quantile if year==2000, lcolor(green) lpattern(##-) ) ///  
			(scatter IHME_LEfemale quantile if year==1990, mcolor(blue) msymbol(Th) ) ///  
			(lfit IHME_LEfemale quantile if year==1990, lcolor(blue)  ) ///  
			(lfit IHME_LEfemale quantile if year==2000, lcolor(blue) lpattern(##-) ) ///  
			(scatter IHME_LEfemale quantile if year==2010, mcolor(blue) msymbol(Oh) ) ///  
			(lfit IHME_LEfemale quantile if year==2010, lcolor(blue)   ) ///  
			,  subtitle("(B) IHME life expectancy estimates", size(4)) graphregion(color(white)) ///
			graphregion(color(white)) ylabel(#3)  xtitle("Poverty percentile", height(4)) ytitle("Life expectancy at birth")   ///
			legend(order(1 "Men 1990" 3 "M 2000" 4 "M 2010" 6 "Women 1990" 8 "W 2010" 9 "W 2010") symxsize(5) symysize(.2) row(2) size(3)) saving(LE1IHME, replace) nodraw
			
			grc1leg   LE1.gph LE1IHME.gph, graphregion(color(white)) legendfrom("LE1.gph") saving("FigS5", replace)
			
	
		*************************************************************************************************
		*  Figure S6: Male 3-year mortality rates by poverty percentile across age groups, with county groups fixed in 1990 and held constant in 2010  *
		*************************************************************************************************
		
			use ${data_cleaned}LE_quantile, clear
			drop if year==2000
			keep if q1990==1

			twoway ///
			(scatter LEmale quantile if year==1990, mcolor(green) msymbol(T) ) ///  
			(lfit LEmale quantile if year==1990, lcolor(green) ) ///  
			(scatter LEmale quantile if year==2010, mcolor(green) msymbol(h) ) ///  
			(lfit LEmale quantile if year==2010, lcolor(green) ) ///  
			(scatter LEfemale quantile if year==1990, mcolor(blue) msymbol(Th) ) ///  
			(lfit LEfemale quantile if year==1990, lcolor(blue)  ) ///  
			(scatter LEfemale quantile if year==2010, mcolor(blue) msymbol(Oh) ) ///  
			(lfit LEfemale quantile if year==2010, lcolor(blue)   ) ///  
			,  subtitle("(A) Level", size(4)) graphregion(color(white)) ///
			graphregion(color(white)) ylabel(#3)  xtitle("Poverty percentile in 1990", height(4)) ytitle("Life expectancy at birth")   ///
			 legend(order(1 "Men 1990" 3 "Men 2010" 5 "Women 1990" 7 "Women 2010") symxsize(3) symysize(.2) row(2) size(3.5)) saving(LE1, replace) nodraw
			
			*absolute change
			use ${data_cleaned}LE_quantile, clear
			drop if year==2000
			keep if q1990==1

			bysort quantile  (year): gen dmale=(LEmale[_n+1]-LEmale) if year==1990
			bysort quantile  (year): gen dfemale=(LEfemale[_n+1]-LEfemale)  if year==1990
					
			twoway ///
			(scatter dmale quant , mcolor(green) msymbol(S) ) ///  
			(lfit dmale quant if year==1990, lcolor(green) ) ///  
			(scatter dfemale quant , mcolor(blue) msymbol(Sh) ) ///  
			(lfit dfemale quant if year==1990, lcolor(blue) ) ///  
			, subtitle("(B) Change", size(4)) graphregion(color(white)) ylabel(#3)  xtitle("Poverty percentile in 1990", height(4)) ytitle("Change in life expectancy at birth")   ///
			saving(LE2, replace)  ///
			 legend(order(1 "Change Men" 3 "Change Women") symxsize(3) symysize(.2) row(2) size(3.5)) nodraw
			
			graph combine LE1.gph LE2.gph, graphregion(color(white)) saving("FigS6", replace)
		
				reg dmale quant 
				reg dfemale quant 
				
				
		*************************************************************************************************
		*  Table S1: Population and economic characteristics of county groups ordered by poverty rate, in 1990 and 2010.*
		*************************************************************************************************
		
			use ${data_cleaned}mort_quantiles_county, clear
			keep if qtype=="Poverty quantile" & q1990==0

			replace pop=pop/1000000
					collapse poverty incMed incpc (rawsum) pop [aw=pop], by(quant year)
		
		
					drop if year==2000
					sort quant year
					foreach var of varlist pop poverty incMed incpc{
					by quant: gen `var'2010=`var'[_n+1]
					rename `var' `var'1990
					}
					keep if year==1990
					drop year
					
					order quant population1990 population2010 poverty1990 poverty2010 incMedianHH1990 incMedianHH2010 incpc1990 incpc2010
					brow
					


		**************************************************************************************************
		*  Table S2: Life expectancy at birth of county groups ordered by poverty rate, in 1990 and 2010.*
		**************************************************************************************************

			use ${data_cleaned}LE_quantile, clear
			keep if q1990==0 & year!=2000
			keep year quant LEmale SEmale LEfemale SEfemale
			
			foreach var of varlist LEmale SEmale LEfemale SEfemale{
			gen `var'1990=`var' if year==1990
			bysort quant (year): gen `var'2010=`var'[_n+1] if year==1990
			drop `var'
			}
			
			keep if year==1990
			
			browse quant LEmale19 SEmale19 LEmale20 SEmale20 LEfemale19 SEfemale19 LEfemale20 SEfemale20
		

		******************************************************************************************************************************
		*  Table S3: Male 3-year mortality of bottom and top poverty county group and slope of fitted regression line, 1990 vs. 2010.*
		******************************************************************************************************************************

			* Mortality rates
			use if inlist(quant,5,100) & qtype=="Poverty quantile" & q1990==0 & year!=2000 using ${data_cleaned}mort_quantiles_county, clear
			
			gen mort=(deaths/pop)*1000
			
			gen se=((deaths/pop)*(1-deaths/pop)/pop)^.5*1000
			
		
			gen q5_1990=mort if quan==5 & year==1990
			bysort male  age (quant year): gen q5_2010=mort[_n+1] if quan==5 & year==1990
			bysort male  age (quant year): gen q100_1990=mort[_n+2] if quan==5 & year==1990
			bysort male  age (quant year): gen q100_2010=mort[_n+3] if quan==5 & year==1990
			
			gen se5_1990=se if quan==5 & year==1990
			bysort male  age (quant year): gen se5_2010=se[_n+1] if quan==5 & year==1990
			bysort male  age (quant year): gen se100_1990=se[_n+2] if quan==5 & year==1990
			bysort male  age (quant year): gen se100_2010=se[_n+3] if quan==5 & year==1990
			
			keep if quan==5 & year==1990
			gsort -male age
			brow male age q5_1990 se5_1990 q5_2010 se5_2010 q100_1990 se100_1990 q100_2010 se100_2010
			
			
			* Regression slopes
			use if qtype=="Poverty quantile" & q1990==0 & year!=2000 using ${data_cleaned}mort_quantiles_county, clear
			
			gen mort=(deaths/pop)*1000
			
			qui foreach g of numlist  0 1 {
			foreach a of numlist  0 2 7 12 17 22 27 32 37 42 47 52 57 62 67 72 77 82 90{

			xi: reg mort i.year*quantile if age==`a' &  male==`g'
			estimates store reg`g'_`a'
			 }
			 }
			
			********************** Regression estimates *************************
			*Slope of fitted regression line in 1990: "quantile"                *
			*Slope of fitted regression line in 2010: "quantile"+"_IyeaXq~2010" *
			*p-value of difference: p-value of "p-value of difference"          *
            *********************************************************************
			
			*males
			estimates table reg1_0 reg1_2 reg1_7 reg1_12 reg1_17 reg1_22 reg1_27 reg1_32 reg1_37 reg1_42 reg1_47 reg1_52 reg1_57 reg1_62 reg1_67 reg1_72 reg1_77 reg1_82 reg1_90, p
			*females
			estimates table reg0_0 reg0_2 reg0_7 reg0_12 reg0_17 reg0_22 reg0_27 reg0_32 reg0_37 reg0_42 reg0_47 reg0_52 reg0_57 reg0_62 reg0_67 reg0_72 reg0_77 reg0_82 reg0_90, p
				
	
			
